

(研究ノート)

テスト処理—操作と実行

テスト処理のための BASIC 言語 によるプログラミング（2）

難　波　俊　裕

統一テストを処理するためのプログラムを製作したことは貴重な経験であった。その過程で知り得た技術を利用すれば、文献ファイルとその検索、単語検索などのプログラムの可能性が射程圏内に見えてきたからである。ここに発表するクラス単位のテスト処理プログラムは統一テスト処理プログラムの応用というより、むしろ、その縮小版ないし再編集にすぎない。しかし、前回の発表は中間発表であったうえ、構想と構成を中心に述べざるを得ない事情があったから、操作と実行を示すことで、それを補完する意味もあろうと思う。

去る五月、南山大学で催された大学英語教育研究会中部支部大会で「カード・リーダーのためのプログラム」を中心に発表した際、午前の最初の時間帯が割り当てられた。内容の特殊性から聴衆が少なかろうと判断した主催の責任者の配慮に、筆者も同感であったのだが、当日その時刻には教室が一杯になったのには、一寸と驚いた。しかし、プログラミングを自ら実践している英語教師は数名を出なかったようである。プログラムの実行の例示は用意していなかったので、紹介の中心であった編成の工夫は大多数が理解できなかつたであろうと懸念する。

その反省もあって、ここでは焦点を操作と実行に置いて述べる。このプログラムは、パソ・コンに関心がなくても、教育の現場で楽に採点できれば、テストによる学習意欲が喚起できるのではなかろうかと思っている教

師の日頃の要求に応えるものである。プログラミングに習熟した者には、プログラム自体から、その構成、操作、実行は容易に想像できるから、本稿の大部分は冗漫な解説と映るであろうが、意図は、プログラムの紹介ではなく、むしろ、BASIC プログラミングを知らない人でも利用できるよう、そのためのマニュアルをめざしたことを見頭に読んで欲しい。

1. 0. 概要

最大100名のクラスに最大40題の5肢選択の設問を課し、「一覧表」「個人成績票」「分布線グラフ」「S P 分析表」が20~30分以内に印刷されるよう、2個のプログラムを用意した。

使用したパソ・コン・システムは前回と同様である。16ビット・パソ・コン：PC-9801 (NEC), 16ドット・プリンター：NK3618-22 (NEC), カード・リーダー：DCD56B (安立電気), 8インチ・ディスク・ユニット：PC-9881K (NEC), システム・フォーマット：PC-98H87(K) がそれである。

本論に入るまえに、カード・リーダーとその周辺について知り得たことを概説しておく。

1. 1. カード・リーダ DCD56B

DCD56B の特性は最適マーク・シートが60欄であることである。既に『教養論叢』Vol. 25, No. 1 で述べたように、80欄カードが全く読み取れないというのではなく、記入者が学生などの如く、不特定しかも多数である場合は、記入上の注意が徹底しないから、DCD56B が誤読に陥り易く、80欄カードでは能率が阻害された。誤読の生じたカードは、その度に、修正しなければならなかったからである。

この機械は読み込んだ光信号を電気信号としてコンピューターに送り込むだけである。それを数字信号に変換するのはホスト・コンピューターの解析プログラムに依らなくてはならない。光信号を読み込む性能は1分間170枚で、充分な性能を持っているが、BASIC の解析プログラムでは、数字

信号への変換は1分間12ないし14枚を超えることは難しいらしい。これを機械語に改めれば、変換スピードは、どんなに小さくみても、10倍になる。しかし、機械語アセンブラーを組むためにはそのニーモニック (Mnemonic Code) を学ばなくてはならないし、変換機能をもつ、コンパイラ (Compiler) も市販されてはいるが、価格が高い上に、OSソフトに対する知識が乏しい現状では安易に購入するわけにもゆかない。

一方、カード・リーダー側で信号を数字に解析した上で、コンピューターに送信する高級機もある。例えば、本学の計算機室に設置されている MELCOM M521-C1 がそうで、1分間に読み取る速度も250枚と大きいが、性能に相応しく DCD56 系の 5, 6 台分に相当する価格である。計算機室では、いくら性能がよくても英語科からは遠く、しかも、学生の演習にも使用されているので、彼等の作業を中断させることになるから、日常の利用には難点がある。やはり安くてそれなりの性能と信頼性を有する DCD56 系の機械は、利用に便利なところに置けることを考慮すると、充分有用である。

1. 2. 60欄マーク・シート

80欄マーク・シートは「中京大学標準マーク解答用紙」として既に日比野省三教授が開発されている。今回設計した60欄マーク・シートは、それを殆どそのまま踏襲した。記して感謝を表わしたい。

このカードは DCD56B 専用というのではなく、MELCOM M521-C1 にも適合するように、学年欄の前にペイス・マークを置いた。この最初のペイス・マークは DCD56B にとっては不要でありダミーになるべきものである。しかし、印刷担当者との連絡が悪く、位置が 3 ミリほど右に寄ったため、このマークが生きた。これを無視することはプログラムの上で簡単に対応できた。

カードと機械の読み取りとの適合は完璧と言ってよいのだが、マニュアルの仕様にもとづいて紙質、インクをそれぞれ「JISC624」「JISD111 ブルー」と指定したにもかかわらず、その通り実行されなかつたようである。「アンリッジDEMO カード」と比較して紙質の色がやや濃くインクの色が異っていて淡すぎ、総体的に不鮮明の出来であった。光線の乏しい机上では記入しにくいのではないかと危ぶまれる。が、ともかく、定期試験に 2 クラス実施したかぎりでは、鉛筆マークが僅かにずれていても、そのため生じた誤読は 1 例もなかつた。

2. 0. プログラム・ロード

統一テスト処理プログラムでは約20個のプログラム群から、処理の段階に応じて必要なものを選択する方法を探ったが、このテスト処理では、2 個のプログラムを順次使用すればよい。“Card” と “Analysis” を装備したディスクを予めシステム・デスクとしておけば、これをディスク・ユニットのドライブ #1 に挿入し、パソ・コンの電源スイッチを入れると「立ち上る」。この時

Disk version

How many files (0-15)? ■

と■が点滅し聞いてくるが、ファイルの必要数は3個以下なので「Return」キーを叩くだけでよい。すると画面(CRT Display)には、その下に

NEC N-88 BASIC (86) Version 1.0
339980 Bytes Free
Ok

の文字が続いて現われる。ここで

RUN "Card"

をキー入力(Key Input)し「Return」キーを叩けば、カード・リーダーによって回答(Response)を読み取り採点するプログラムが塔載(Load)される。

RUN "Analysis"

を入力すれば、既に採点処理されディスクに収納(Output)されたファイルから、そのデータを読み取り(Input)，棒グラフとSP分析表を印刷するプログラムが塔載される。

8吋倍密ディスクでは勿論のこと5吋倍密デスクを使用しても、1組50人前後のデータの占めるトラックは2トラックにすぎないので、プログラム2個と同じディスクにデータを収納しても楽に20回分のテストのデータを収納できる余裕がある。8吋倍密ディスクでは一枚で延べ140組分の試験資料を収納できる。

2. 1. "Card"

先ず、このプログラムで使用する変数記号(Variables)のうち重要なものを出現の順に説明する。変数記号の解説はプログラムの初期設定前のREM文にアルファベット順に挙げているので、重複になるがここで触れておく必要がある。

CODE\$: Class Name (処理の対象となる組を示す。例えば
"Sho : 2 : E" は商学部商学科2年E組を示し、Out

-put, Input の際には File Name として使用する)

- COM:E71 : File Name for Card Reader DCD56B(カード・リーダーを開く際の Code である)
- N : No. of Cards having been Read (既に読み込まれたカード数を示す変数で、後にレコード番号Aとして使用する)
- KK : No. of Questions (設定する設問の数)
- DLN : Down Line Number (必要な欄を示す変数で、最大は47である)
- P\$: Numerals of A\$, Constituents of CR\$ (カード・リーダーからの信号 A\$ を1930行から2070行までで解析し、変換する文字変数)
- YRS : Number for Year Grade (学年を示す文字変数)
- NO\$: STR\$ of NO or Shusseki Bangou (出席番号を示す数字変数 NO を文字変数として変換 (STR\$) したもの。実際は NO\$ が先に得られそれを数字変数に変換する (VAL) 操作をする)
- CR\$: String of Answers (回答 (Responses) をその順序に連結した文字配列である。NO\$ が “99999” のものは正答群として RA\$ で示される)
- R\$: Numrls of 2 Dimensions to point out where Double-Marked (1欄中に二重マークが存在した時、事故 (Accident) として扱うが、それを発見するためライン・プリンターで印刷されるマーク、つまり、“0”から“9”までの符号を表わす文字変数である。R\$ が存在しないときは “.” で示す)
- C : Shusseki Bangou or NO (各学生の出席番号、読み込まれた時は文字変数 NO\$ で示すが、File に格納するときは数字配列Cとして扱う)

- X : Length of RA\$ or CR\$ (RA\$ 即ち CR\$ のバイト数で、常に頭（左端）に空白バイトが付いているので設問数よりも 1 個多い KK+1 を示す)
- K : Allocated Points for Question (各問の配点)
- F\$: Bit of RA\$ or Right Answer of a Question (各問の正答を示す文字変数で、この F\$ は回答 (Response) を示す G\$ と対比され、一致すれば、採点符号 Q\$ が “+” に変換される)
- FR : Sum of Right Answers of each Question among All Students (設問に対する正答がクラス全体でどれだけあったかを示す数字変数)
- G\$: Bit of CR\$ or a Student's Answers (回答 (Response) を示す)
- Q\$: “+” to indicate G\$=F\$ else Q\$=G\$ (採点後の学生の正答を示す文字変数)
- E\$: String of Marked Answers of Students (Q\$ を設問の数だけ繋いだもの。この左端にはやはり空白バイトが存在する。
- O : No. of Right Answers of Students (得点を示す。正答ごとに配点を加え、得られた数を示す数字変数)
- A : Record No or I (読み込んだ順序を示す数字変数で 1 から N までの変数)
- BB : Acquired Points or L (各学生の得点、即ち、L である)
- D : No. of Right Answers or O (各学生の正答数、O を File に収納するとき D(I) として扱う)
- W : Sum Total of Acquired Points by all (組全体の正答数の合計)
- DATE\$: Date of Marking (Y/M/D) (コンピューターに内

蔵する時計から得られる年／月／日の文字配列)

- YMD\$: Year / Month / Day or DATE\$ (採点の年月日として、DATE\$を固定したもの。文字配列)
- AV : Average or W/N (得点の平均)
- STD : Standard Deviation of Class or S/N (組の標準偏差)

2. 2. 読み取りの操作と実行

「Run “Card”」のキー入力に応えて、黄色の文字、

Program : Card Reader

が画面 (CRT, Cathode Ray Tube) 中央に現われ、その下に白色で

What is the CODE of the cards to be treated?

と尋ねてくるが、クラス名をキー入力する。それは例えば「Sho: 2:E」などでよいが、File Name としても使用するものなので、9字以内でなければならぬ制約がある。体育学部には12班があるので、「Tail: 1201」とでもしておけば、此クラスに課した15回目試験は「Tail: 1215」と表わすことができよう。

続いて、同じく白文字で、同じ画面の下に

You have only to make a print of

と現われ、緑色で更に、

the OLD DATA. (Y/N)

と表示されるから、「Y」または「y」を入力すれば、「古い資料」即ち、既に File に収納したデータを読み始める。カード・リーダーを動かそうと思うなら、「N」または「n」を入力する。カード読み取りを選択したとして説明を続ける。

ここでカード・リーダーが開く (l. 1590)。そしてカードを機械に載せ

カード・リーダーのスタート・スイッチを押せと指示がある。

Put the cards on the READER and push 'START' switch.

Strike 'Return Key' if ready? ■

ここで万全であれば、「Return」キーを押せばよい。

続いて、白色の質問

How MANY questions are there? ■

には、単に設問の数を入力すればよい。「Sho: 2: E」の例では「35」である。これが変数 KK となる。

ここからカード・リーダーが始動する。しかし、カードの積み方を述べておかねばならない。

正答を記入したカードは、カードの山の底になるよう積まなければならぬ。学年は記入せず、正答カードの出席番号はかならず “99999” と五行の “9” を記入しなくてはならない。行番号1620が示すように、読み始めのNは0であり、N=0かつ NO\$= “99999” である場合にのみ「正答」即ち RA\$ が成立するからである (l. 2200)。仮に、「正答カード」が学生カードの中に混じっておれば、正答としては登録されず、いかなる正答も存在しないことになる。

始動直後に、読み取った数字の列を表示し始めた時、注意深く観察すれば、

0 99999 211321311.....

の如く、学年を示す数字が表示されないことに気付くであろう。

これは、RA\$ を表示するプログラムは2200行であるのに対して、学生のカード (CR\$) を処理するプログラムは2210行から2260行であることにによる。

カードをよく揃えて載せないと、事故を誘発する。これはカード・リーダーの赤ランプが灯り、空送りするから注意を要する。

順調にカードを読み取っている時の事故発生は、リーダーの停止とともにホスト・コンピューターが警報を鳴らす。その原因は 1) 「出席番号」に記

入洩れがあるか, 同一欄に二個のマーク, 即ちダブル・マークがあるか, 2) 解答欄にダブル・マークがあるかである。原因となったカードは最終のカードで, これは読み取り済のカード群の最上部にあるから, 直ちに抜き取る。

上記二種のカードは該当する部分を修正しなければ, 正常な採点ができない。修正すべき個所の発見と確認に必要なのが二次元のマークの表示であり, これを行うのは Line Printer である。これを実行するのが2270行から2700行のプログラムである。右肩に “Shu Ban” とあれば出席番号が閑門で阻止されたことを示し, “Mrd Ans” とあれば解答欄のどこかに修正すべきダブル・マークがあることを示す。

右の例では, “53309” であるべき出席番号の下2桁が同一欄に記入されている。“211321…” に始まる数列と二次元表示の数字が一致しないのは, このカードの CR\$ はこの時点では, 出来上っておらず, 現在の CR\$ は一枚前の CR\$ を表示するからである。

全てのカードの読み取りが完了すると, 1900行と1910行が実行される。これは事故を知らせる警

報と同じものであるが, その際には, 単に「Return」キーを叩くだけでよかつたのに対し, 読み取り完了の時には, “CLOSE” または “close” を入力しなければならない。これによってカード・リーダーを閉じるのである。

1分12ないし14枚の速度でしか読み取らないので, 50枚程度で, ここまでで, 数分を要する。

3. 3. 採点の操作と実行

採点の操作は配点の入力で始まる。黄色文字が

How many Points do you allocate for each question?

ShuBan

2	5	3	3	2	1	3	1	1	-	1	3	2	2	1	3	4	1	2	3	2	1	2	3	1	
0	0	1	1	1	1	1	1	1	1	1	1	1	1	1	1	1	1	1	1	1	1
2	2	2	2	2	2	2	2	2	2	2	2	2	2	2	2	2
3	3	3	3	3	3	3	3	3	3	3	3	3	3	3	3	3
5	5	5	5	5	5	5	5	5	5	5	5	5	5	5	5	5
9	9	9	9	9	9	9	9	9	9	9	9	9	9	9	9	9

不良カードの例

No. 1 : Pts ? ■

と尋ねるから、「1番」の配点を入力すると「2番」の配点を尋ね、設問ごとに任意の点数を配する方法を採用した。全部の配点が終了したところで、黄色文字が終り白色文字が

Are you sure (Y/N) ? ■

ときくから、改めたいと思えば “N” を入力すると、配点操作を最初からやりなおすことができる。“Y” を入力すれば正答を表示するから各問の正答が再確認できる。それに対する

Strike ‘Rtrn Key’ if ready. ■

に対しては、否ということはできない。が、「Return」キーを押すまではプログラムは進行しないから、いくら念入りに検討してもよい。

採点中は入力順を示すレコード番号と出席番号とが白字であらわれ、その後に採点結果を示す E\$ と正答数 (O) と得点 (L) とが黄色で表示される。終了は警報が鳴り、画面が暗転し、黄色の

The Number of the Students (Sho: 2: E) 49

が現われ、白字で例によって、

Strike ‘Retrn Key’ if ready. ■

が、「Return」キーを押してくれと求めるから、それに応ずると、出席番号順に File を並べかえる (Sort) 作業が始まる。すると、

Now sorting

Now Sorting

と、配列変換が実行されていることを示す文字がN—1回出現する。この終了は緑色の

COMPLETED ××××× Now List is Printing.

と白色の

Do you Keep Data into the File? (Y/N)? ■

が同時にあらわれるから、Disk にファイルがまだ収納されていない時は、
“Y”を入力しなければならない。既に Disk にファイルが存在していればこの必要はなく、 “N” を入力すればよい。

“Y”を入力すれば Line Print する前に Subroutine “Keep Data in Seq. File” (ll. 4220--4290) を済ませてから、Line Printer へ信号を送り、「一覧表」を打ち出す。

CLASS NAME: Sho:2:E		STUDENTS: 49	AV: 38.24	STD: 7.06	DATE: 84/09/29	
Right Answers		211321311-1322313213123411314232312		T.P	R.A	
21	53213	++2++3+++3++3++3++2+23+1+41421-+5++		41	19	
14	53214	++2++++322++112+1+++4+412213+1-4533		28	14	
37	53216	+++++3+++3211+12+++4111+2++34-+5+5		38	18	
44	53219	3+32+2+3+1++++++3++14+4+2++455+		44	21	
28	53221	++2++++2+2+222++21+1+4+++++14+		46	22	
25	53227	++2++2+23+231+1+++++2341234+13-4533		28	14	
15	53229	++2++3++2++1+2+++++34134+4+2+-455+		39	19	
11	53232	++++++1+++1++++2+2++3+342++-3++4		50	24	
10	53237	++31+312+1++31+++++31++344+33-4++		39	18	
48	53239	++2+++++33+23+12++1314+424++3-3+5+		36	17	
30	53242	++3++3+++3++3+2+2++2143+21224++5+5		41	19	
5	53248	++3233+++3++++++21+23132+-3544		34	17	
6	53251	++2++3+223+21+13++++2+1+34+2++-++++		44	20	
26	53252	++++1+++222+3+2+1+31311+232+---555		30	15	
16	53253	++3133+2+1++312++++243++2+12+3-4555		28	14	
4	53258	++++++1++1++332+2++24114+2+23--5++		41	19	
49	53260	++3++2++2+21+++++21+++4++1++155		48	23	
31	53261	3+21331232+233+++++1342124132+-322+		19	9	
39	53266	++++++21+2++++++4++321+2++-5++		53	25	
41	53267	+++++3+2+1++++2+2+1+1+14243++1455+		43	21	
42	53271	++2++3+++2++1+1++++3+414+4+2++455+		44	21	
46	53277	++32+2+3+3++++++3++14+4+2+-4534		40	20	
19	53279	++2++31+33++1+2+1+++343+24423-4545		28	14	
20	53286	++++++21+11+++++44133+1++-3251		40	20	
7	53289	+2+++++33+++3132+++3112+21+14--543		33	16	
45	53291	+2++3++31++1+132++4+1432414-4+5+		34	16	
17	53294	++2++++331+13++21+2+2341231+++4155		28	14	
18	53295	++2++3++23++3+++++52+55322+34-45++		38	18	
2	53296	++2++3++22+++++1+++34++++++4253		46	23	
9	53297	+++++1++1++2+2+2+241+4+2+24--++		49	23	
3	53300	+23+13+222-2---+2+3+41244+---555		25	12	
24	53302	+22+12++32++3+13++243+1221++1-+52+		32	15	
29	53303	+222322++1++12+++2+4113231+++15+-		32	15	
13	53305	+++++1+21++3+++++1++41341+2+--+24		44	21	
43	53306	++2+13++3++++++2211++4+4++4352+		45	22	
38	53308	++++++2+3++312++++2+2+++++++-++		60	28	
1	53309	++2+13+223++3++1+++3121+413+1-+5+5		34	16	
36	53311	++++++1++2++++12++++3++23+++45++		52	25	
8	53313	++323312+2++32+1++23++2+++--35+5		37	18	
33	53314	1+2+++++3+2+1+++++14321241+---12+		37	18	
47	53316	++2++213+12+3+1++++++1++++++5+		53	25	
22	53320	322++31+32+2+3+21+++31234224+3-+531		23	11	
23	53323	+++++3+2+3++32++++3++1+4+4++-455+		47	23	
12	53325	+22+13+223+21+21++31+1324+++-3533		26	13	
35	53326	++2+331222++3+1++2231413+2213-+535		23	11	
32	53327	++32+3++3++311+++224++4+4++-254		39	19	
27	53329	++2++3++3++31++1+++4+12+43+1-3555		36	18	
34	53331	++3+33++23+2+31++++43+122++3+-4++		41	19	
40	53337	++2233+++1++++++24+32+4+3+13243		38	19	

Allo. Pnt 22222222022222222222222223333 Total: 1874 Fulmark: 73
901

その「一覧表」とは別に、各問の正答数を示す「表2」が画面 (CRT)

1984. 11

テスト処理——操作と実行 テスト処理のための
BASIC 言語によるプログラミング(2) (難波)

203 (825)

とプリンターとに同時に作製される。それが終れば（正確にはプリンターへ送信が終れば）

Do you want another copy? Yes (1)/ No (0)? ■

が、「もう一枚」同じ物がいるかと尋ねるから「0」を入力すると、青色の“Now it begins to print ‘Individual Slit’”の表示があって個人票の印字が始まる。

FREQUENCY of Right Answers for Each Question											
Class: Sha:2:E		Year/Month/Day: 84/09/29,		N.of Questions: 35							
1, , , , 2, , , , 3, , , , 4, , , , 5, , , , 6, , , , 7, , , , 8, , , , 9, , , , 10											
10	1	45(2)	42(2)	14(2)	39(2)	35(2)	16(2)	36(2)	31(2)	28(2)	0(0)
10	1	44(2)	36(2)	22(2)	29(2)	22(2)	37(2)	35(2)	48(2)	41(2)	38(2)
20	1	8(2)	12(2)	23(2)	16(2)	15(2)	14(2)	13(2)	34(2)	27(2)	29(2)
30	1	5(3)	18(3)	10(3)	16(3)	23(3)	0()	0()	0()	0()	0()
Sum total of Points: 1874, Sum Total of R.A.: 901											
N.of Students: 49		MAXIM.: 60		MINIM. 19							
Average : 38.24		Standard D : 7.06									

正 答 頻 度 表

36, 37, 38, 39, 40間に相当する欄が“0()”であるのは問題が存在せず、従って、正答は「0」で配点も存在しないから「()」である。第10問は正答数零配点零“0 (0)”を示す。個人成績票は切り離して短柵の形で学生に返却し、講評するためのものである。49名程度を対象とし、16ドット印字で10分足らずの時間を必要とする。8ドット印字でなら、もっと速くなるが、読み得るが粗雑になることは避けられない。

STUDENTS: 49		AV: 38.24	STD: 7.06	DATE: 84/09/29
Right Answers	211321311-1322313213123411314232312		T.P	R.A
1	53213	++2++3+++3++3++3++2+23+1+41421-+5++	41	19
All. Points: 222222220222222222222222222233333				Ful Mark 73

STUDENTS: 49		AV: 38.24	STD: 7.06	DATE: 84/09/29
Right Answers	211321311-1322313213123411314232312		T.P	R.A
2	53214	++2++++322++112+1+++4+412213+1-4533	28	14
All. Points: 222222220222222222222222222233333				Ful Mark 73

個 人 成 績 票

一際が完了すれば、黄色の

Program : CARD READER is OVER

To go to Program : Analysis, first hit 'Retrn Key'.

に続き白文字の

Then, when 'OK' sign has appeared, input 'run "Analysis"', hitting 'Retrn'.

が現われるから、その通り実行すれば、プログラムは、"Analysis" に変わる。

2. 5. Disk File からの実行

2.2 の「読み取り操作と実行」で、緑色の

the OLD Data. (Y/N)? ■

の間があった (l. 1515)。これが分岐点で、"Y" を入力すると、既に作られた File を Fl. Disk から CODE\$ を手掛にして、そのデータを入力する。カード・リーダーの読み取り、配点、採点、配列を飛越して3530行からの印刷から実行する。既成のデータを単に「一覧表」「個人成績票」として利用するのであれば、この飛び越し操作を行えばよい。

3. 0. フロー・チャート

上記のプログラムの構成を図示すれば次の頁のとおりである。

4. 1. 分布グラフの操作と実行


先ず「Run "Analysis"」をキー入力により実行すれば、画面 (CRT) 中央に黄色の次の表示が現われる。

Program : Analysis

その下に白色の文字で

Code \$? ■

と、File Name を要求するから、例えば「Sho: 2: E」をキー入力すると、プログラム "Card" で Disk #1 に収納した順編成ファイル (Sequential File) からの入力 (Input) が始まる。同時に、その内容を表示する



フロー・チャート : Program Card

が、画面に收まり切らないので、実行が順調に行われていることを意味するにすぎない。

次に、緑色で画面中央に


Full Mark : 73

と浮かび、その下に白色の文字

Hit 'Retrn Key' if ready. ■

が出現するから、「Return」キーを叩けば、プログラムは分布を示す棒グラフを描く段階に進行する (1. 1640)。

暗転すると、ほとんど瞬時に次の表があらわれ、同時に、同じ図表を描くべく Line Printer が音を立て始める。



まだ印刷が終了しないのに、今度は黄色の

Hit 'Retrn Key' if ready to go to SP Analysis. ■

が表の下に現われるが (1. 2070), 印刷に必要な信号は全て Line Printer が受信しているので、印刷の終了を待つまでもなく、「Return」キーを叩けばよい。すると画面の表は消え、

Now working, No 1

Now working, No 2

Now working, No 3

の文字が青色で現われる。このとき、コンピューターは G(J, 1), G(J, 2), G(J, 3), G(J, 4), G(J, 5), G(J, 6) の二次元の座標をもつ 6 種, 40 間で 240 個の記号へ分類しているのである。この作業はカードのマーク読み取りの解析作業と同種なので、1 個の CARD\$, つまり、1 名の資料を解析するために 2 秒近くの時間を必要とする。

4. 2. S P 分析表の実行

解析が終了すると、白色で結果を表示するが、画面からはみ出してしまうので、順調に進行していることを知る以外には意味がない。S P 分析表はプリンターが専ら打ち出す。

CLASS:Sho:2:E (84/09/29) Students: 49, AV: 38.24

R.A.	I	1	2	3	4	5	-1	I	1	2	3	4	5	-
1) 2	I	1	45	3	0	0	01	491	2.0	91.8	6.1	0.0	0.0	0.0
2) 1	I	42	7	0	0	0	01	491	85.7	14.3	0.0	0.0	0.0	0.0
3) 1	I	14	24	11	0	0	01	491	28.6	49.0	22.4	0.0	0.0	0.0
4) 3	I	3	7	39	0	0	01	491	6.1	14.3	79.6	0.0	0.0	0.0
5) 2	I	6	35	8	0	0	01	491	12.2	71.4	16.3	0.0	0.0	0.0
6) 1	I	16	6	27	0	0	01	491	32.7	12.2	55.1	0.0	0.0	0.0
7) 3	I	11	2	36	0	0	01	491	22.4	4.1	73.5	0.0	0.0	0.0
8) 1	I	31	13	5	0	0	01	491	63.3	26.5	10.2	0.0	0.0	0.0
9) 1	I	28	13	8	0	0	01	491	57.1	26.5	16.3	0.0	0.0	0.0
10) -	I	15	14	20	0	0	01	491	30.6	28.6	40.8	0.0	0.0	0.0
11) 1	I	44	4	0	0	0	11	491	89.8	8.2	0.0	0.0	0.0	2.0
12) 3	I	2	11	36	0	0	01	491	4.1	22.4	73.5	0.0	0.0	0.0
13) 2	I	9	22	17	0	0	11	491	18.4	44.9	34.7	0.0	0.0	2.0
14) 2	I	11	29	9	0	0	01	491	22.4	59.2	18.4	0.0	0.0	0.0
15) 3	I	12	14	22	0	0	01	491	24.5	28.6	44.9	0.0	0.0	2.0
16) 1	I	37	7	4	0	0	01	491	75.5	14.3	8.2	0.0	0.0	2.0
17) 3	I	11	3	35	0	0	01	491	22.4	6.1	71.4	0.0	0.0	0.0
18) 2	I	0	48	1	0	0	01	491	0.0	98.0	2.0	0.0	0.0	0.0
19) 1	I	41	7	1	0	0	01	491	83.7	14.3	2.0	0.0	0.0	0.0
20) 3	I	5	5	38	0	1	01	491	10.2	10.2	77.6	0.0	2.0	0.0
21) 1	I	8	17	16	8	0	01	491	16.3	34.7	32.7	16.3	0.0	0.0
22) 2	I	16	12	10	11	0	01	491	32.7	24.5	20.4	22.4	0.0	0.0
23) 3	I	11	4	23	10	1	01	491	22.4	8.2	46.9	20.4	2.0	0.0
24) 4	I	23	1	8	16	1	01	491	46.9	2.0	16.3	32.7	2.0	0.0
25) 1	I	15	15	8	11	0	01	491	30.6	30.6	16.3	22.4	0.0	0.0
26) 1	I	14	14	8	13	0	01	491	28.6	28.6	16.3	26.5	0.0	0.0
27) 3	I	15	7	13	14	0	01	491	30.6	14.3	26.5	28.6	0.0	0.0
28) 1	I	34	5	6	4	0	01	491	69.4	10.2	12.2	8.2	0.0	0.0
29) 4	I	4	13	5	27	0	01	491	8.2	26.5	10.2	55.1	0.0	0.0
30) 2	I	6	22	8	6	0	01	491	12.2	59.2	16.3	12.2	0.0	0.0
31) 3	I	2	0	5	1	0	411	491	4.1	0.0	10.2	2.0	0.0	83.7
32) 2	I	1	18	10	17	0	31	491	2.0	36.7	20.4	34.7	0.0	6.1
33) 3	I	4	5	10	0	30	01	491	8.2	10.2	20.4	0.0	61.2	0.0
34) 1	I	16	5	6	5	17	01	491	32.7	10.2	12.2	10.2	34.7	0.0
35) 2	I	2	23	6	5	12	11	491	4.1	46.9	12.2	10.2	24.5	2.0

S P 分析表のなかで、左側の欄の正答頻度、右側の欄のその百分率を示す数字の下に下線を施すよう工夫した。従って、いちいち正答に照合しないでも正答の他の回答に対する割合を知ることができる。この下線は Line Printer に内蔵されている機能を活用したものである (II. 2580, 2590; II. 2650, 2660)。

プログラムの終了とともに、

You have done with Program : Analysis

を黄色で中央に表示する。その下に白色の字が

To get out of Program, just strike 'Retrn Key'.

を告げるから、例のとおり「Return」キーを叩けば電源を切ってよい状態になる。

プログラム “Analysis” の所用時間は対象50名位いで3～4分である。“Card” の始めから “Analysis” の終了まで、例示の “Sho : 2 : E” で言えば、約17分を要した。

勿論、ここで続けて別な資料を解析したいときには、プログラム自体は装備された状態 (Load) なので、“Run” をキー入力し、「Return」キーを叩くか、単に「f.5」のキーを叩くかすれば、この項の始めに示した “Program : Analysis” 以下が画面 (CRT) 中央に出現する。その後の作業は既に述べたとおりである。


S P 分析表に示される数を、より視覚に訴える形、1設問について、例えば、6本の棒グラフ、あるいは、扇形、円グラフに表示することも考えられる。統一テストのように、2,000名の単位の円グラフなら、その百分比の変動は、問題の適性、対象としての学生の質の変化など示唆する手掛りとはなるが、数十人を対象とする比率は1名の反応 (Response) の影響が大きく、そのため百分比が変りやすい。クラス単位では誰が、どの設問に失敗したか、個人に関心をむける方が重要であろう。

また円グラフをコンピューターの画面に描くのは簡単であるが、それを Line Printer で打ち出すと、縦長の墮円に変形されるので、正常の円グ

ラフを印刷するには高級な技術が必要である。詳わしくは述べないが、それやこれやで、このプログラムでは、割愛した。

コンピューター採点においては S P 分析表が得られたところから、本来の教師の役割が始まる。個々の設間に取りあげた教授細目をどう導入すればよいか、特定の学生群の受容にそれは相応しいか、学生の特性をどう捉えるべきか、問題の提示の仕方など、本質的な問題提起とその対策は出題した教師の内部において、資料との対話から始まることである。

4. 3. フロー・チャート



5. メモリの負荷

扱う対象が小数であるから “Card”, “Analysis” の両プログラムとも、その実行に当って、負荷のメモリーは予想どおり少かった。拡張ラム・ボード1枚を載せたこのシステムでは、配列宣言 (Statement of Dimension) を600、つまり1学部全員用に拡大することも可能である。

メモリの消費を検証しておく。実行時にプログラムを中断して次の命令を実施すれば4個の数字の応答が得られる。

```
for I=0 to 3: print fre (I); : next
```

`fre(0)` は「メモリ・エリア」の大きさ、`fre(1)` は「テキスト・エリア」の基模、つまりプログラムを収納する容量、`fre(2)` は両者の合計の容量で、`fre(3)` が「シンボル・テーブル・エリア」別名「配列エリア」の容量で、表示の数字は正確には残量を示す。

立ち上げ時の容量と使用時との各エリアの空白 (free) 容量、つまり残量とを比較すれば、プログラムの負荷が計算できる。

	無 荷	Card (負荷)	Analysis (負荷)
<code>fre (0)</code>	63,208	58,341 (4,967)	60,802 (2,394)
<code>fre (1)</code>	53,524	40,224 (13,300)	46,943 (6,581)
<code>fre (2)</code>	116,732	98,565 (18,167)	107,745 (8,987)
<code>fre (3)</code>	223,248	218,784 (4,464)	220,624 (2,624)
<code>fre(2)+fre(3)</code>	339,980	317,349 (22,631)	328,369 (11,611)

プログラム負荷が14Kバイト、配列容量負荷が5Kバイトにすぎない。使用したコマンドも特殊なものはないので、8ビット系パソコンでも利用できるのではなかろうか。

6. プログラム

プログラム 1 の 1

```

1000 REM-----CARD READER PROGRAM "Class Unit" V. 2.0 -----
1010 '
1020 '
1030 '      Card Reader Program for Class Unit (File Nme "Card")
1040 '                                         Version 2.0 by Toshihiro NAMBA
1050 '                                         Date of Make: Sept 25, 1984
1060 ' Variants:--
1070 '     A: Record No. or I
1080 '     A$: Input #1 from DCD56B
1090 '     AA$: String of F$s or Right Answers
1100 '     AV: Average or W/N
1110 '     B: Year Grade or YR
1120 '     BB: Acquired Points or L
1130 '     C: Shusseki Bangou or NO
1140 '     CODE$: Class Name
1150 '     CR$: String of Answers
1160 '     D: No. of Right Answers or 0
1170 '     D$: CR$ of Students
1180 '     DATE$: Date of Marking (Y/M/D)
1190 '     E$: String of Marked Answers of Students
1200 '     F$: Bit of RA$ or Right Answer of a Question
1210 '     FR: Sum of Right Answers of a Question among All Students
1220 '     FFR$: String of STR$ of All FR's
1230 '     G$: Bit of a Student's Answers
1240 '     K: Allocated Points for Question
1250 '     K$: STR$ of K or 1 Byte while K in 2 Bytes
1260 '     KK: No. of Questions
1270 '     L: Acquired Points of a Student
1280 '     N: No. of Cards having been Read
1290 '     NO$: STR$ of NO or Shusseki Bangou
1300 '     O: No. of Right Answers of a Student
1310 '     P$: Numerals of A$, Constituents of CR$
1320 '     R$: Numrals of 2 Dimensions to point out Where Double-Marked
1330 '     RA$: CR$ of Right Answers or of Card "99999"
1340 '     Q$: "+" to indicate G$=F$ else Q$=G$
1350 '     S: Sum of SQR(SS)s or Sum of N times STD
1360 '     SS: 2nd Power of Deviation or (AV-L)^2
1370 '     STD: Standard Deviation of Class or S/N
1380 '     Y: Sum Total of Right Ansers by All
1390 '     YMD$: Year/Month/Day or DATE$
1400 '     W: Sum Total of Acquired Points by All
1410 '
1420 '
1430 '
1440 '
1450 CLEAR,&H1E00
1460 CLS:WIDTH 80,25:CONSOLE 0,25,0,1:LOCATE 10,10:COLOR 6
1470 LT$="PROGRAM : CARD READER":X$=AKCNV$(LT$):PRINT X$
1475 PRINT TAB(15); "for MARKING a class of STUDENTS. ":"PRINT:PRINT
1480 COLOR 7:PRINT TAB(10); " What is the CODE of the cards to be treated";
1485 INPUT CODE$
1490 DIM P$(80),A(100),B(100),C(100),D$(100),E$(100),BB(100),K(50)
1500 DIM FR(80),K$(80)

```

訂正： 1060 ' Variants → 1060 ' Variables

1450 CLEAR, &H1E00 → 1450 CLEAR, &H6000

プログラム 1 の 2

```

1510 LOCATE 10,20:PRINT "You have only to make a print of"
1515 LOCATE 10,21:COLOR 4:INPUT "the OLD Data. (Y/N)"; ZA$ :COLOR 7
1520 IF ZA$="Y" OR ZA$="y" GOTO 4320:PRINT :PRINT
1530 PRINT "*****"
1540 PRINT "*** CARD READER PROGRAM ***"
1550 PRINT "*** for ONE Class (";CODE$; ")";TAB(51); "***"
1560 PRINT "***"
1570 PRINT "***"
1580 DC1$=CHR$(&H11):STX$=CHR$(&H2):ETX$=CHR$(&H3):NAK$=CHR$(&H15)
1590 OPEN"COM:E71"AS#1
1600 BEEP:PRINT "Put the cards on the READER and push 'START' switch"
1610 INPUT "Strike 'Rtrn key' if ready";S$ 'Dummy Input
1620 N=0
1630 INPUT "How MANY questions are there";KK
1640 *AGAIN
1650 BEEP:COLOR 1:PRINT "***Now it began reading. NO.";
1660 COLOR 2:PRINT N;:COLOR 1:PRINT "sheet" :COLOR 7
1670 PRINT #1,DC1$;
1680 A$=INPUT$(1,#1)
1690 IF A$<>STX$ THEN PRINT "Redo from the start.":STOP
1700 A$=INPUT$(1,#1)
1710 IF A$=NAK$ THEN *ERROR.
1720 WH$="":WL$=""
1730 *LOOP.
1740 WH$=WH$+A$
1750 A$=INPUT$(1,#1)
1760 IF A$=ETX$ THEN *QUIT.
1770 WL$=WL$+A$
1780 A$=INPUT$(1,#1)
1790 IF A$<>ETX$ THEN *LOOP.
1800 *QUIT.
1820 DLN=KK+7
1830 GOSUB *DECIPHER LEN(WH$)
1840 GOSUB *RECORD
1850 N=N+1
1860 ERASE P$
1870 DIM P$(80)
1880 GOTO *AGAIN
1890 *ERROR.
1900 BEEP:BEEP:PRINT "***Accident in reading*****"
1910 INPUT "Strike RETURN if you are to stay. Else input 'CLOSE' if you want
to go out.":W$
1920 IF W$="CLOSE" OR W$="close" THEN CLOSE #1 :GOTO 2740 ELSE *AGAIN
1930 *DECIPHER
1940 FOR I=1 TO DLN
1950 SUB$=MID$(WH$,I,1):SUB2$=MID$(WL$,I,1)
1960 IF SUB$=CHR$(68) THEN IF SUB2$=CHR$(64) THEN P$(I)="0"
1970 IF SUB$=CHR$(72) THEN IF SUB2$=CHR$(64) THEN P$(I)="1"
1980 IF SUB$=CHR$(80) THEN IF SUB2$=CHR$(64) THEN P$(I)="2"
1990 IF SUB$=CHR$(96) THEN IF SUB2$=CHR$(64) THEN P$(I)="3"
2000 IF SUB$=CHR$(64) THEN IF SUB2$=CHR$(65) THEN P$(I)="4"
2010 IF SUB$=CHR$(64) THEN IF SUB2$=CHR$(66) THEN P$(I)="5"
2020 IF SUB$=CHR$(64) THEN IF SUB2$=CHR$(68) THEN P$(I)="6"
2030 IF SUB$=CHR$(64) THEN IF SUB2$=CHR$(72) THEN P$(I)="7"
2040 IF SUB$=CHR$(64) THEN IF SUB2$=CHR$(80) THEN P$(I)="8"
2050 IF SUB$=CHR$(64) THEN IF SUB2$=CHR$(96) THEN P$(I)="9"
2060 IF SUB$=CHR$(64) THEN IF SUB2$=CHR$(64) THEN P$(I)="-"

```

(注) 1940行以下は 600名を対象とした試験的走行においては、次のように別の解析プログラムを用意した。これによると二重マークの修正の手間がはぶける。

```

1920 FOR I=1 TO DLN
1925 DIM P%(10)
1930 W=ASC(MID$ WH$,I,1):IF W >64 THEN GOSUB *CAT.1
1940 W2=ASC(MID$ WL$,I,1):IF W2 >64 THEN GOSUB *CAT.2
1941 FOR J=0 TO 9
1942 IF P%(J) THEN P$(I)=RIGHT$(STR$(P%(J)-1),1):MM=MM+1
1943 NEXT J
1945 IF MM>1 THEN P$(I)=""
1950 IF P$(I)="" THEN P$(I)="-"
1955 ERASE P%:MM=0
1960 NEXT I
1970 GOTO 2210
1980 *CAT.1
1990 W=W\4
2000 IF W MOD 2 THEN P%(0)=1
2010 W=W\2
2020 IF W MOD 2 THEN P%(1)=2
2030 W=W\2

```

プログラム 1 の 3

```

2070 NEXT I
2080 RETURN
2090 *RECORD
2100 RC=N
2110 YRS=P$(2)
2120 NO$=""
2130 FOR I=3 TO 7: NO$=NO$+P$(I):NEXT I: NO$=VAL(NO$): NO$=STR$(NO)
2140 IF LEN(NO$)<>6 THEN PRINT "ERROR!!!*****in Shusseki Bangou*****": A$="ShuBa
n":GOTO *CHECK
2150 CR$="" "
2160 FOR I=8 TO DLN: CR$=CR$+P$(I):NEXT I: IF LEN(CR$)<DLN-6 THEN PRINT "ERROR!!
!*****in Marks of Answers *****": A$="MrdAns":GOTO *CHECK
2170 ----- Right Answer -----
2180 ,
2190 ,
2200 IF N=0 AND NO$=" 99999" THEN RA$=CR$:PRINT N;NO$:RA$:RETURN
2210 -----Recording Students' answers-----
2220 PRINT USING "###";RC;
2230 B(N)=VAL(YRS):PRINT " ";B(N);
2240 C(N)=VAL(NO$):PRINT C(N);
2250 D$(N)=CR$:PRINT D$(N)
2260 RETURN
2270 *CHECK
2280 DIM R$(60,12)
2290 FOR I=1 TO DLN
2300   W=ASC(MID$(WH$,I,1)):IF W THEN GOSUB *PAT.1
2310   W=ASC(MID$(WL$,I,1)):IF W THEN GOSUB *PAT.2
2320 NEXT I
2330 GOTO 2600
2340 *PAT.1
2350 IF W MOD 2 THEN R$(I,0)="R"
2360 W=W¥2
2370 IF W MOD 2 THEN R$(I,1)="X"
2380 W=W¥2
2390 IF W MOD 2 THEN R$(I,2)="O"
2400 W=W¥2
2410 IF W MOD 2 THEN R$(I,3)="1"
2420 W=W¥2
2430 IF W MOD 2 THEN R$(I,4)="2"
2440 W=W¥2
2450 IF W MOD 2 THEN R$(I,5)="3"
2460 RETURN
2470 *PAT.2
2480 IF W MOD 2 THEN R$(I,6)="4"
2490 W=W¥2
2500 IF W MOD 2 THEN R$(I,7)="5"
2510 W=W¥2
2520 IF W MOD 2 THEN R$(I,8)="6"
2530 W=W¥2
2540 IF W MOD 2 THEN R$(I,9)="7"
2550 W=W¥2
2560 IF W MOD 2 THEN R$(I,10)="8"
2570 W=W¥2
2580 IF W MOD 2 THEN R$(I,11)="9"
2590 RETURN
2600 REM ----- CHECK PRINT -----
2610 LPRINT YR$;NO$;CR$;" ";A$

```

```

2040 IF W MOD 2           THEN P%(2)=3
2050   W=W¥2
2060 IF W MOD 2           THEN P%(3)=4
2070 RETURN
2080 *CAT.2
2090 IF W2 MOD 2          THEN P%(4)=5
2100   W2=W2¥2
2110 IF W2 MOD 2          THEN P%(5)=6
2120   W2=W2¥2
2130 IF W2 MOD 2          THEN P%(6)=7
2140   W2=W2¥2
2150 IF W2 MOD 2          THEN P%(7)=8
2160   W2=W2¥2
2170 IF W2 MOD 2          THEN P%(8)=9
2180   W2=W2¥2
2190 IF W2 MOD 2          THEN P%(9)=10
2200 RETURN

```

プログラム 1 の 4

```

2620 FOR J=2 TO 11
2630   FOR I= 1 TO DLN
2640     IF R$(I,J)="" THEN LPRINT CHR$(165);
2650     LPRINT R$(I,J);
2660   NEXT I
2670   LPRINT
2680 NEXT J
2690 FOR Z=1 TO 5:LPRINT : NEXT Z
2700 ERASE R$ : GOTO 1910
2710 '
2720 '
2730 '
2740 CLS
2750 PRINT "*****"
2760 PRINT "***"
2770 PRINT "**           PROGRAM FOR MARKING          **"
2780 PRINT "**           Students' CODE(";CODE$;");TAB(53);**"
2790 PRINT "*****"
2800 X=LEN(RA$) :PRINT :PRINT:
2810 COLOR 6:PRINT "How many Points do you allocate for each question?"
2820 ALL.PNT$=""
2830 FOR J=2 TO X
2840   PRINT "No. ";J-1;" : Pts";:INPUT K(J)
2850   K$(J)=RIGHT$(STR$(K(J)),1): ALL.PNT$=ALL.PNT$+K$(J)
2860 NEXT J
2870 COLOR 7:INPUT "Are you sure? (Y/N)";A$
2880 IF A$="Y" OR A$="y" THEN ELSE 2810
2890 ,
2900 '
2910 DIM F$(80)
2920 PRINT RA$:AA$=""
2930 FOR H=2 TO X:F$(H)=MID$(RA$,H,1):AA$=AA$+F$(H):NEXT H
2940 LINE INPUT "Strike 'Rtrn Key' if ready";A$
2950 PRINT -----
2960 ,
2970 REM -----FILE READING -----
2980 N=N-1:FOR J=2 TO X:FR(J)=0:NEXT
2990 FOR I=1 TO N
3000   PRINT I;TAB(10);C(I);
3020 DIM G$(60),Q$(60)
3030   E$(I)=""
3040   FOR J=2 TO X
3050     G$(J)=MID$(D$(I),J,1)
3060     IF G$(J)=F$(J) THEN Q$(J)="+":FR(J)=FR(J)+1:O=O+1:L=L+K(J) ELSE
3070     E$(I)=E$(I)+Q$(J)
3080   NEXT J
3090   COLOR 6: PRINT TAB(16); E$(I);      ";
3100   PRINT USING "###";O;
3110   PRINT USING "###";L;
3120   PRINT :COLOR 7
3130   A(I)=I:BB(1)=L:D(I)=0
3140   W=W+L:Y=Y+O
3150 ERASE G$,Q$:O=0:L=0
3160 NEXT I
3170 BEEP:BEEP:PRINT CHR$(12)
3180 REM-----Sorting -----

```

プログラム 1 の 5

```

3190 PRINT "*****"
3200 PRINT "** ARRANGING IN ORDER OF Shusseki Bangou **"
3210 PRINT "*****"
3220 COLOR 6:PRINT "The Number of the Students (";CODE$;"");N :COLOR 7
3230 LINE INPUT "Strike Rtrn Key, if ready";A$
3240 FOR M=N-1 TO 1 STEP -1
3250 PRINT "Now sorting"
3260 FOR J=1 TO M
3270 IF C(J)>C(J+1) THEN SWAP C(J),C(J+1):SWAP A(J),A(J+1):SWAP E$(J),
E$(J+1):SWAP BB(J),BB(J+1):SWAP D(J),D(J+1)
3280 NEXT J
3290 NEXT M
3300 BEEP:BEEP:CLS:LOCATE 10,10
3310 YMD$=DATE$
3320 COLOR 4:PRINT "COMPLETED*****Now LIST is printing.":CLS:LOCATE 10,10
3330 COLOR 7:INPUT "Do you keep Data into the file? (Y/N)";XA$
3340 IF XA$="y" OR XA$="Y" THEN GOSUB 4210 ELSE 3360
3350 GOSUB *FREQUENCY
3360 GOSUB *STATISTIC
3370 *PRINT.           'for class data
3380 LPRINT "      CLASS NAME: ";CODE$
3390   LPRINT "STUDENTS:";:LPRINT USING"####";N;
3400   LPRINT "      AV:"; :LPRINT USING"##.##";AV;
3410   LPRINT "      STD:";:LPRINT USING"##.##";ST;
3420   LPRINT "      DATE:";YMD$
3430 FOR I= 1 TO 70: LPRINT CHR$(148)::NEXT:LPRINT
3440 LPRINT "Right Answers    ";AA$; "      T.P      R.A"
3450 FOR I=1 TO 70: LPRINT CHR$(149)::NEXT: LPRINT
3455 MAX=0:MINI=FULMRK
3460 FOR I=1 TO N
3465 IF MAX<BB(I) THEN MAX=BB(I)
3466 IF MINI>BB(I) THEN MINI=BB(I)
3470   LPRINT USING "###";A(I);
3480   LPRINT TAB(7);C(I);
3490   LPRINT TAB(16);E$(I);";    ";
3500   LPRINT USING"###";BB(I);
3510   LPRINT USING"###";D(I)
3520 NEXT I
3530 FOR I=1 TO 80:LPRINT CHR$(148)::NEXT:LPRINT
3540 LPRINT "All o. Pnt ";:LPRINT TAB(16);ALL.PNT$::LPRINT TAB(65);"Fulmark: ";
3545 LPRINT USING"###";FULMRK
3550 LPRINT TAB(50);"Total: ";
3560 LPRINT USING"#####";W::LPRINT "    ";:LPRINT USING"#####";Y
3570 LPRINT CHR$(&HC):CLS
3580 PRINT "FREQUENCY of R.A. for Each Q"
3590 PRINT TAB(10);CODE$;
3600 PRINT TAB(30);YMD$
3610 PRINT TAB(4);"      1,,2,,3,,4,,5,,6,,7,,8,,9,,10"
3620 FOR I=1 TO 80:PRINT CHR$(149)::NEXT:PRINT
3630 PRINT TAB(4);CHR$(150)::FOR J=2 TO 11:PRINT USING"##";FR(J)::PRINT "(";K$(J);")";:NEXT:PRINT
3640 PRINT "10";:PRINT TAB(4);CHR$(150);
3650 FOR J= 12 TO 21: PRINT USING"##";FR(J)::PRINT "(";K$(J);")";:NEXT:PRINT
3660 PRINT "20";:PRINT TAB(4);CHR$(150);
3670 FOR J= 22 TO 31: PRINT USING"##";FR(J)::PRINT "(";K$(J);")";:NEXT:PRINT

```

プログラム 1 の 6

```

3680 PRINT "30";:PRINT TAB(4);CHR$(150);
3690 FOR J= 32 TO 41: PRINT USING"####";FR(J);:PRINT "(";K$(J);")";:NEXT:PRINT
3700 FOR I=1 TO 70:PRINT CHR$(148);:NEXT:PRINT
3710 PRINT "Sum Total of Points:";:PRINT USING"#####";W;
3720 PRINT ", Sum Total of R.A.:";:PRINT USING"#####";Y
3730 PRINT "N.of Students:";:PRINT USING"####";N:
3740 PRINT " Average :" ;:PRINT USING"####.#";AV
3750 PRINT " Standard D :" ;:PRINT USING"####.#";ST
3755 PRINT " Maxim. :" ;:PRINT USING"####";MAX
3756 PRINT " Minin. :" ;:PRINT USING"####";MINI
3760 LPRINT "FREQUENCY of Right Answers for Each Question"
3770 LPRINT TAB(5);"Class: ";CODE$;
3780 LPRINT TAB(30);"Year/Month/Day: ";YMD$;
3790 LPRINT ", N.of Questions: ";KK
3800 LPRINT TAB(5);" 1, , , , 2, , , , 3, , , , 4, , , , 5, , , , 6, , , , 7, , , , 8, , , , 9, , , , 10"
3810 FOR I=1 TO 80:LPRINT CHR$(149);:NEXT:LPRINT
3820 LPRINT TAB(5);CHR$(150);:FOR J=2 TO 11:LPRINT USING"###";FR(J);:LPRINT "(";K$(J);")";:NEXT:LPRINT
3830 LPRINT "10";:LPRINT TAB(5);CHR$(150);
3840 FOR J= 12 TO 21:LPRINT USING"###";FR(J);:LPRINT "(";K$(J);")";:NEXT:LPRINT
3850 LPRINT "20";:LPRINT TAB(5);CHR$(150);
3860 FOR J= 22 TO 31:LPRINT USING"###";FR(J);:LPRINT "(";K$(J);")";:NEXT:LPRINT
3870 LPRINT "30";:LPRINT TAB(5);CHR$(150);
3880 FOR J= 32 TO 41:LPRINT USING"###";FR(J);:LPRINT "(";K$(J);")";:NEXT:LPRINT
3890 FOR I=1 TO 80:LPRINT CHR$(148);:NEXT:LPRINT
3900 LPRINT "Sum total of Points:";:LPRINT USING"#####";W;
3910 LPRINT ", Sum Total of R.A.:";:LPRINT USING"#####";Y
3920 LPRINT "N.of Students:";:LPRINT USING"###";N;:LPRINT TAB(25);" MAXIM.:";
3925 LPRINT USING"###";MAX;:LPRINT TAB(45);" MINIM.:";:LPRINT USING"###";MINI
3930 LPRINT " Average :" ;:LPRINT USING"###.#";AV
3940 LPRINT " Standard D :" ;:LPRINT USING"###.#";ST
3950 LPRINT CHR$(&HC)
3960 INPUT "Do you want another copy? Yes(1)/No(0)";XZ$
3970 IF XZ$="1" THEN GOTO *PRINT. ELSE CLS:LOCATE 10,10:COLOR 5
3980 PRINT "Now it begins to print 'Individual Slit'."
3990 LOCATE 10,12:LINE INPUT "Hit 'Retrn Key' if ready.>";W$
3995 COLOR 7
4000 *PRINT2.
4010 '
4020 FOR I=1 TO N
4030   LPRINT "STUDENTS:";:LPRINT USING"#####";N;
4040   LPRINT " AV:"; :LPRINT USING"##.#";AV;
4050   LPRINT " STD:";:LPRINT USING"##.#";ST;
4060   LPRINT " DATE: ";YMD$;
4070 FOR J= 1 TO 70: LPRINT CHR$(148);:NEXT:LPRINT
4080 LPRINT "Right Answers ";AA$; " T.P R.A"
4090 FOR J=1 TO 70:LPRINT CHR$(149);:NEXT:LPRINT
4100   LPRINT USING "###";I;
4110   LPRINT TAB(7);C(I);
4120   LPRINT TAB(16);E$(I);"; ";
4130   LPRINT USING"###";BB(I);
4140   LPRINT USING"###";D(I)
4150 FOR J=1 TO 70:LPRINT CHR$(148);:NEXT:LPRINT
4160 LPRINT "All. Points:";:LPRINT TAB(16);ALL.PNT$;
4165 LPRINT TAB(62);"Full Mark";:LPRINT USING"###";FULMRK
4170 LPRINT CHR$(&HA);CHR$(&HA)
4180 NEXT I

```

プログラム 1 の 7

```

4190 LPRINT CHR$(&HC) :CLS
4191 LT$="Program: CARD READER is OVER.":X$=AKCNV$(LT$)
4192 LOCATE 10,10:COLOR 6:PRINT X$
4193 LOCATE 10,13:PRINT "To go to Program:Analysis,first hit 'Retrn Key'."
4194 COLOR 7:PRINT TAB(10);"Then, when 'Ok' sign has appeared,"
4195 PRINT TAB(10);"input 'run 'Analysis'' ',hitting 'Retrn'." 
4200 END
4201 '
4210 '-----Subroutines-----
4211 '
4220 '----- Keep Data in Seq. File -----
4230 GOSUB 4440
4240 OPEN "1:"+CODE$ FOR OUTPUT AS #2
4250 WRITE #2,YMD$,N ,AA$,FFR$,X,KK
4260 FOR I=1 TO N:WRITE #2,A(I),C(I),E$(I),BB(I),D(I):NEXT
4270 FOR J=2 TO X:WRITE #2,FR(J),K(J):NEXT
4280 CLOSE #2
4290 RETURN
4300 '
4310 '
4320 '----- Take Data from Seq. file -----
4330 '
4340 OPEN "1:"+CODE$ FOR INPUT AS #2
4350 INPUT #2,YMD$,N ,AA$,FFR$,X ,KK
4360 FOR I=1 TO N:INPUT #2,A(I),C(I),E$(I),BB(I),D(I):NEXT
4370 FOR J=2 TO X:INPUT #2,FR(J),K(J):NEXT
4380 CLOSE #2 :FOR I=1 TO N: W=W+BB(I):Y=Y+D(I):NEXT:FULMRK=0:ALL.PNT$=""
4390 FOR J=2 TO X: K$(J)=RIGHT$(STR$(K(J)),1)
4395 FULMRK=FULMRK+K(J):ALL.PNT$=ALL.PNT$+K$(J):NEXT
4400 GOTO 3360
4410 '
4420 *FREQUENCY
4430 '
4440 FFR$="";FULMRK=0:FOR I=2 TO X:FFR$=FFR$+STR$(FR(I)):FULMRK=FULMRK+K(I):NEXT
4450 RETURN
4460 '
4470 *STATISTIC
4480 REM-----STATISTICS-----
4490 AV=W/N
4500 FOR I= 1 TO N:SS=(AV-BB(I))^2:S=S+SQR(SS): NEXT I :ST=S/N
4510 PRINT TAB(5);"Total of Pnts";W;" Date:";DATE$ 
4520 PRINT "STUDENTS:";N;"." AVE: " ;AV;" STD:";ST
4530 RETURN

```

プログラム 2の1

```

1000 REM-----Program for Analysis-----
1010 '
1020 '
1030 '      Test Analysis Program
1040 '                      Version 1.0      by Toshihiro NAMBA
1050 '                      Date of Make: Sept 30, 1984
1060 '      Variants:--
1070 '          AA$: String of Right Answers or RA$
1080 '          ALL.PNT: Allotted Point or K
1090 '          B$: Bit of AA$ or Right Answers
1100 '          AV: Average of All Students' Points or W/N
1110 '          B$: Bit of AA$ or Right Answers
1120 '          C$: Bit of CARd$
1130 '          CARD$: String of Marked Answers by Students or E$
1140 '          CODE$: Class Name
1150 '          F: Frequency of a Mount of Points at a Level J
1160 '          FULMRK: Full Mark of Test
1170 '          G: Number of Markings of Each Response for a Question
1180 '          J: a Dimension of G or Number of Questions
1190 '          KK: Number of Questions on Key-Input
1200 '          L: Points for a Step
1210 '          MAX: Maximum of Points
1220 '          MINI: Minimum of Points
1230 '          N: Number of Cards
1240 '          NO: Shusseki Bangou
1250 '          POINTS: Points Obtained by Stutents
1260 '          RECD: Record No. or Order of having read Cards
1270 '          RIGANS: Number of Right Answers by Students
1280 '          S:
1290 '          SS:
1300 '          SUM: Sum Total of Right Answers for a Question or FR
1310 '          T: Total of Right Answers
1320 '          TT: Total of G's or Markings of each Response i.e. N
1330 '          W: Total of Points
1340 '          X: Number of Questions or Length of AA$ or i.e. KK
1350 '          YMD$: Date of Treatment or Year/Month/Day
1360 '
1370 '
1380 '
1390 CLEAR,&H1E00
1400 CLS:W10TH 80,25:CONSOLE 0,25,0,1:
1410 LT$="Program: Analysis":X$=AKCNV$(LT$):COLOR 6:LOCATE 10,10:PRINT X$
1420 COLOR 7:LOCATE 15,12:INPUT "Code$";CODE$
1430 DIM RECD(100),NO(100),CARD$(100),POINTS(100),RIGANS(100)
1440 DIM F(20),SUM(80),ALL.PNT(50)
1450 '-----INPUT OF DATA from S. File on Disk #1-----
1460 OPEN "1:"+CODE$ FOR INPUT AS #1
1470 INPUT #1,YMD$,N, AA$,FRR$,X,KK
1480 PRINT YMD$,N,AA$
1490 W=0:T=0
1500 FOR I=1 TO N
1510   INPUT #1,RECD(I),NO(I),CARD$(I),POINTS(I),RIGANS(I)
1520   PRINT RECD(I),NO(I);CARD$(I);POINTS(I);RIGANS(I)
1530   W=W+POINTS(I):T=T+RIGANS(I)
1540 NEXT I
1550 FOR J=2 TO X:INPUT #1,SUM(J),ALL.PNT(J):FULMRK=FULMRK+ALL.PNT(J):NEXT
1560 CLOSE

```

訂正： 1060 ' Variants → 1060 ' Variables

1390 CLEAR, &H1E00 → 1390 CLEAR, &H6000

プログラム 2 の 2

```

1570 '----- GRAPH BY 20 LEVELS-----
1580 '
1590 '
1600 GOSUB *STATISTIC
1610 FOR I=0 TO 19: F(J)=0 :NEXT
1620 '
1630 CLS:LUCATE 10,10:COLOR 4:PRINT "Full Mark: ";FULMRK:COLOR 7
1640 LOCATE 10,12:LINE INPUT "Hit 'Rtrn Key' if ready.":W$
1650 L=FULMRK/20 :MAX=0:MINI=FULMRK
1660 FOR I=1 TO N
1670     J=INT(POINTS(I)/L):F(J)=F(J)+3
1680     IF POINTS(I)>MAX THEN MAX=POINTS(I)
1690     IF POINTS(I)<MINI THEN MINI=POINTS(I)
1700 NEXT I
1710 CLS:
1720 PRINT TAB(5);;"CLASS: ";CODE$;"(";YMD$;")";;" Students:";N;" AV: ";AV
1730 FOR I=1 TO 70:PRINT CHR$(149);:NEXT
1740 PRINT
1750 FOR J=19 TO 0 STEP -1
1760     PRINT USING"####";INT(J*L)+1;:PRINT CHR$(151 );
1770     IF F(J)=0 THEN GOTO 1800
1780     FOR K=1 TO F(J): PRINT CHR$(134);:NEXT K
1790     PRINT TAB(55);USING"##";F(J)/3;
1800     PRINT
1810 NEXT J
1820 PRINT "Maxi: ";MAX;," Minim: ";MINI;
1830 PRINT TAB(55);;"One Step:";:PRINT USING"##.###";L
1840 LINE INPUT"Hit 'Retrn Key' if ready.":W$
1850 LPRINT TAB(5);;"Frequency Graph of Obtained Points":LPRINT
1860 LPRINT TAB(5);;"CLASS: ";CODE$;"(";YMD$;")";;" Students:";N;" AV: ";
1870 LPRINT USING"##.##";AV
1880 FOR I=1 TO 70:LPRINT CHR$(149);:NEXT
1890 LPRINT
1900 FOR J=19 TO 0 STEP -1
1910     LPRINT USING"####";INT(J*L)+1;:LPRINT CHR$(151 );
1920     IF F(J)=0 THEN GOTO 1950
1930     FOR K=1 TO F(J): LPRINT CHR$(135);:NEXT K
1940 LPRINT TAB(65);USING"##";F(J)/3;
1950     LPRINT
1960 NEXT J
1970 FOR I=1 TO 5:LPRINT CHR$(149);:NEXT:LPRINT CHR$(151);
1980 FOR I=1 TO 22
1990     FOR J=1 TO 2:LPRINT CHR$(149);:NEXT J
2000     LPRINT CHR$(151);
2010 NEXT I
2020 LPRINT
2030 LPRINT TAB(5);;"0"           5          10          5
20"           5
2040 LPRINT
2050 LPRINT "Maxim: ";MAX;," Minim: ";MINI;
2060 LPRINT TAB(55);;"One Step:";:LPRINT USING"##.###";L
2070 COLOR 6:LINE INPUT "Hit 'Rtrn Key' if ready to go to SP Analysis.":W$ :COLO
R 7

```

プログラム 2 の 3

```

2080 LPRINT CHR$(&HC)
2090 '-----SP ANALYSIS -----
2100 ERASE RECD,NO,RIGANS,F,SUM
2110 DIM G(42,7),B$(50)
2120 CLS :COLOR 5
2130 FOR I=1 TO N
2140 DIM C$(60)
2150 X=LEN(AA$)
2160 PRINT TAB(30); "Now working, NO." ;I
2170     FOR J= 1 TO X
2180         B$(J)=MID$(AA$,J,1)
2190         C$(J)=MID$(CARD$(I),J,1)
2200         IF B$(J)="1" AND C$(J)="+" THEN G(J,1)=G(J,1)+1
2210         IF B$(J)="2" AND C$(J)="+" THEN G(J,2)=G(J,2)+1
2220         IF B$(J)="3" AND C$(J)="+" THEN G(J,3)=G(J,3)+1
2230         IF B$(J)="4" AND C$(J)="+" THEN G(J,4)=G(J,4)+1
2240         IF B$(J)="5" AND C$(J)="+" THEN G(J,5)=G(J,5)+1
2250         IF B$(J)="--" AND C$(J)="+" THEN G(J,6)=G(J,6)+1
2260             IF C$(J)="1" THEN G(J,1)=G(J,1)+1
2270             IF C$(J)="2" THEN G(J,2)=G(J,2)+1
2280             IF C$(J)="3" THEN G(J,3)=G(J,3)+1
2290             IF C$(J)="4" THEN G(J,4)=G(J,4)+1
2300             IF C$(J)="5" THEN G(J,5)=G(J,5)+1
2310             IF C$(J)="--" THEN G(J,6)=G(J,6)+1
2320     NEXT J
2330 ERASE C$
2340 NEXT I
2350 COLOR 7
2360 '----- Table of SP(length ; Questions:width;Answers)-----
2370 FOR I=1 TO 70: PRINT CHR$(149);:NEXT:PRINT
2380 FOR J=1 TO X
2390     PRINT J;" ";B$(J);:PRINT TAB(13);CHR$(150);
2400     FOR I=1 TO 6
2410         PRINT USING"####";G(J,I);
2420     NEXT I
2430 PRINT
2440 NEXT J
2450 '-----PRINTING SP TABLE -----
2460 LPRINT "CLASS:";CODE$;" (";YMD$;"") Students:";:LPRINT USING"####";N;
2470 LPRINT ", AV: ";:LPRINT USING"###.##";AV :LPRINT
2480 FOR I=1 TO 75:LPRINT CHR$(149);:NEXT:LPRINT
2490 LPRINT TAB(5);"R.A.";:LPRINT TAB(11);CHR$(150);
2500 LPRINT "    1   2   3   4   5   -";CHR$(150);
2510 LPRINT TAB(41);CHR$(150);"    1   2   3   4   5   -"
2520 FOR I=1 TO 75: LPRINT CHR$(149);:NEXT:LPRINT
2530     TT=0
2540 FOR J=1 TO X
2550     LPRINT USING"###";J;:LPRINT " ) ";B$(J);:LPRINT TAB(11);CHR$(150);
2560     FOR I=1 TO 6
2570         IF I<>VAL(B$(J)) THEN LPRINT USING"###";G(J,I);:GOTO 2600
2580         LPRINT CHR$(27);CHR$(&H58);:LPRINT USING"###";G(J,I);
2590         LPRINT CHR$(27);CHR$(&H59);
2600         TT=TT+G(J,I)
2610     NEXT I
2620     LPRINT CHR$(150);:LPRINT USING"###";TT;:LPRINT CHR$(150);
2630     FOR I=1 TO 6
2640         IF I<>VAL(B$(J)) THEN LPRINT USING"###.##";(G(J,I)/TT)*100;:GOTO 2670

```

プログラム 2 の 4

```
2650      LPRINT CHR$(27);CHR$(&H58);:LPRINT USING "###.##";(G(J,I)/TT)*100;
2660      LPRINT CHR$(27);CHR$(&H59);
2670  NEXT I
2680      LPRINT
2690 ,
2700      TT=0
2710  NEXT J
2720 FOR I=1 TO 75:LPRINT CHR$(149);:NEXT:LPRINT
2730 ,
2740
2750 LPRINT CHR$(&HC):CLS
2760 LT$="You have done with Program: Analysis.":X$=AKCNV$(LT$):COLOR 4:LOCATE 5
,10:PRINT X$ :COLOR 7
2770 PRINT:PRINT:PRINT TAB(15);"To get out of the program, just strike 'Retrn Ke
y',"
2780 PRINT TAB(25);:LINE INPUT W$:LPRINT CHR$(&HC)
2790 END
2800 '-----Subroutines -----
2810 '
2820 *STATISTIC
2830 AV=W/N
2840 FOR I=1 TO N:SS=(AV-POINTS(I))^2:S=S+SQR(SS):NEXT :ST=S/N
2850 RETURN
```